
The Delphi Application Framework library contains two components: TDynBPLServer and TDynBPLClient both based on TDynBPLBase. DynBPL stands for Dynamic Borland Package Library.

In this Tutorial you will learn how easy it is to integrate new functionality to your programs, make them dynamic for future issues and seamlessly deliver new modules to your customers without caring about side effects with the older ones.

If you have detailed questions about the classes / members, please see the online help. 

Create your projects and the project group

( Please notice, that the whole projects are already created, you don’t have to do this by yourself. However, this tutorial will help you to get a better start using the components.

The Main Application

First, create a new project and save it as MainAppUnit.pas respectively MainApp,dpr. We will now create a module container with only one task: To load or unload modules with real functionality.  

Important: In the options of the  project you must define “With runtime packages” and you have to add ALL (in the host) used packages (e.g. vcl,rtl,appfw) in the Edit field beneath.

Add the following controls to the main form:

· A main menu, with menus File / Exit and Modules / Configure and Modules / - (disabled line)

· A TDynBPLServer component, setting Services to “Add.Form”, “Add.Menu”

· A panel, set align to alClient

For now, we only add the functionality for common things and add menu entries to the main menu. Ok, let’s begin.

The exit menu item click (very large…)

procedure TMainForm.exit1Click(Sender: TObject);

begin


Close;

end;

Form Create and Destroy

In the Create procedure we set the infile name of the bpl server to our app name with an ini extension. Then we load the modules that were loaded the last session and finally call a service to inform all loaded modules about the fact, all modules are loaded and initialized and the App will start now. This could be important for packages which are depending on other ones.

procedure TMainForm.FormCreate(Sender: TObject);

begin


DynBPLServer.IniFileName :=



ChangeFileExt( Application.ExeName, '.ini' );


DynBPLServer.LoadFromIniFile;


DynBPLServer.CallService('App.Start',nil,true);

end;

The destroy procedure just informs the packages about the App will end.

procedure TMainForm.FormDestroy(Sender: TObject);

begin


DynBPLServer.CallService('App.End',nil,true);

end;

AppFW Configuration

The ClickEvent of the MenuItem Configure will call the BPL Server’s Configuration Dialog and then save the configuration into the inifile:

procedure TMainForm.Configure1Click(Sender: TObject);

begin


DynBPLServer.ConfigDialog;


DynBPLServer.SaveToIniFile;

end;

Handling Services

Now it’s time to fullfill our promise to handle the published services (you remember this property ?). For the moment we’ll only implement the service to add menu items (Add.Menu). For this reason, we expect Subject as a TPopupMenu instance and add it to our Modules Menu. To assure not getting in trouble we use RTTI to proof and cast the subject. Finally, we set the caption of the passed Popup menu items property to the caption of the appropriate BPL-Client to indicate these menu items belong to it. Notice, that TpopupMenu.items is ONE TMenuItem object which offers the caption property.

By the way, we do not have to care of the life time of the menu, since the parent of the sender has created it (Sender is BPLClient, its parent is the form). When the package will be removed from the application, the menu items will be freed and so automatically removed from the menu bar (thank you, vcl).

procedure TMainForm.DynBPLServerServiceCall(Sender: TObject;

  ServiceText: String; ServiceNumber: Integer; Subject: TObject;


out handled: Boolean);

var


menu: TPopupMenu;

begin


case ServiceNumber of



0:;  // Add.Form



1:  // Add.Menu




if Subject is TPopupMenu then




begin





menu := (Subject as TPopupMenu);





ModulesMenu.Add(menu.Items);





menu.Items.Caption := (Sender as TDynBPLBase).Caption;





end;


end;

end;

OK, this should be enough for now – we have a complete package container. Let’s continue with the first module to get a little bit life into our application.

The first module

Add a new package project with a right-click on the project Group, first select “Add new project”, then Package. Save the project group and the package in the same directory as the MainApp. While package1 is activated, add a new Form from the File Menü. Save the new unit as ExplorerUnit.pas.

Important: In the options of the package you must define “Runtime package” and you should give a short description, since this is shown by the default BPL Server configuration dialog. Also, you should set the output path (and dcp output path) to your project path in order to create the packages there, rather than in projects/bpl.

( When compiling the package, the compiler adds automatically all required packages to the required list. Just confirm the dialog when you first compile the package.

Change the name of the new form to “ExplorerForm”. Now add the following controls to it:

· A popup Menu with as many subitems as you want.
For every item create a click event where you show a message or other silly things

· A TDynBPLClient, where you set
- Caption to “ExplorerClient”
- Description to “This is the Explorer Client from AppFW Tutorial”
- Services to “App.Start” and “App.End”
This should suffice for the first task.

How to create and destroy the form

Important: The right place to create and destroy your form is the initialization respectively the finalization section. Remember, there is no Application creating it for you.

initialization


ExplorerForm := TExplorerForm.Create(nil);

finalization


ExplorerForm.Free;

Send the client menu to the host

Since we’ve now drawn ourself out of the mud, we will go right through the Create event. From here we will send our menu to the host:

procedure TExplorerForm.FormCreate(Sender: TObject);

begin


DynBPLClient1.CallService('Add.Menu', PopupMenu1, false );

end;

Yes, that’s all. Now run your main aplication and select the menu item “Modules / Configure”. Click on the Add button and select “Package1”, then close the dialog. As you will see, your menu items are shown as a submenu for the newly created menu item “Explorer Client”.

Adding forms to the host

To make our form more realistic, we’ll create a tiny explorer on it (Since not everyone has installed the ShellControls Demo package, we’ll create a dummy explorer). Load a suitable ExplorerForm icon and add the following controls to to the form:

· A TreeView, set align to alLeft and add a few nodes

· A horizontal splitter

· A ListView, set align to alClient, ViewStyle to vsReport and add some columns and items.

Then add the following line to the FormCreate Event:


DynBPLClient1.CallService('Add.Form',self,false);

Change to the Main Application and add a Menu “Forms” to the MainMenu and a MenuItem “None” which is checked by default. Add then a private procedure

procedure AddForm(Client: TDynBPLBase; ClientForm: TForm);

to the form and complete the first servce in the CallService event of the BPL Server:


case ServiceNumber of



0:  // Add.Form




if Subject is TForm then





AddForm( (Sender as TDynBPLBase), (Subject as TForm) );

Create an OnClick event for the menu item Forms / None, which is usable for all new form menu items (we create it instantly after this). Here we check the right menu item and set the corresponding form visible.

procedure TMainForm.FormMenuClick(Sender: TObject);

var


i: integer;


mItem: TMenuItem;

begin


mItem := Sender as TMenuItem;


if mItem.Checked then



exit;


for i:=0 to FormsMenu.Count -1 do


begin



FormsMenu.Items[i].Checked := false;



if FormsMenu.Items[i].Tag <> 0 then




TForm( FormsMenu.Items[i].Tag ).Visible := false;


end;


mItem.Checked := true;


if mItem.Tag = 0 then



exit;


TForm( mItem.Tag ).Visible := true;

end;

In the Server’s AddForm procedure we change the Form’s proprties to our needs, most important: parent to Panel1 and visible=false. Then we create a new menu item (again with the sender as owner) and fearlessly save the Form address in the tag to use it later for access. Finally we set the OnClick event and add the menu item to our form menu.

procedure TMainForm.AddForm(Client: TDynBPLBase; ClientForm: TForm);

var


mItem: TMenuItem;

begin


ClientForm.Parent := Panel1;


ClientForm.Visible := false;


ClientForm.Align := alClient;


ClientForm.BorderStyle := bsNone;


ClientForm.BorderIcons := [];


ClientForm.TabStop := true;


ClientForm.TabOrder := 0;


mItem := TMenuItem.Create(Client);


mItem.Caption := ClientForm.Caption;


mItem.tag := integer(ClientForm);


mItem.OnClick := FormMenuClick;


FormsMenu.Add(mItem); 

end;

Now we have a little trouble, because there are two instances which cares about freeing the form. First, the package which will free it in the finalization (you remember ?) and the second: The panel, which is now the parent of the form. In order to work around this mess we add some code to the MainForm.Destroy event and it should look as below:

procedure TMainForm.FormDestroy(Sender: TObject);

var


i: integer;

begin


for i:=0 to FormsMenu.Count -1 do


begin



if FormsMenu.Items[i].Tag <> 0 then



begin




TForm( FormsMenu.Items[i].Tag ).Visible := false;




TForm( FormsMenu.Items[i].Tag ).Parent := nil;



end;


end;


DynBPLServer.CallService('App.End',nil,true);

end;

As you see, we have ended any parental relationship to all the forms, so we can’t get in any trouble. Not before we’ve set the visibility to false, to avoid screen flickering.

( Run your main application and show the explorer form. Go to the configure dialog, unload the package and see how the menu and form will be removed. Load the package again.

Other Objects

OK, now it would be nice to show the client’s form icon with the menu item for the user’s convenience. Therefore we add an ImageList to the main form and extend our AddForm procedure as follow:

Insert:


if not ClientForm.Icon.Empty then



mItem.ImageIndex := ImageList1.AddIcon(ClientForm.Icon);

before


FormsMenu.Add(mItem);

You may ask, how or when will the added icons in the imagelist be removed. You’re right, the icons will stay until the image list will be destroyed and hence memory will be wasted. You can proof it with a simple message. Add a menu Test to the main menu and a new item “Report number of icons” to the Test menu. Create an OnClick event for the latter and call ShowMessage:

procedure TMainForm.Reportnumberoficons1Click(Sender: TObject);

begin


ShowMessage( Inttostr(ImageList1.Count) + ' Icons' );

end;

When you start your project and remove and add the package a several times, you will see how the number of icons increases. This is a problem, since we  don’t care about the lifetime of the menus and forms – as you remember - to avoid maintenance coding for the clients.

However, sometimes we must be aware of the moment when particular objects will be destroyed. To handle this, we have various possibilities. Here are three:

The first way is to introduce a new service “Remove.Form” or any similar to the hosts service list. Then call this service from the destroy event of the relevant form in the package and the host is able to compare the given form with the menu tags in the forms menu and then to delete the corresponding icon with the imageindex property.

This is easy and quite clear so far, but the disadvantage is, that this method must be defined bevor development and moreover, the host is depending of the client’s behaviour. If the client does not call the service, there’s no chance to catch the gap.

The second way is more reliable and independed. In the AddForm procedure we add the line

FreeNotification(ClientForm);

In the MainForm we override the procedure

procedure Notification(AComponent: TComponent; Operation: TOperation); override;

and implement it as follow:

procedure TMainForm.Notification(AComponent: TComponent;


Operation: TOperation);

var


i: integer;


mItem: TMenuItem;

begin


inherited Notification(AComponent, Operation);


if(AComponent = self)


or(AComponent.Owner = self)


or(Operation <> opRemove)


or(csDestroying
in ComponentState) then



exit;


for i:=0 to FormsMenu.Count -1 do


begin



mItem := FormsMenu.Items[i];



if mItem.Tag = integer(AComponent) then



begin




if mItem.ImageIndex >= 0 then




ImageList1.Delete(mItem.ImageIndex);



mItem.ImageIndex := -1;



end;


end;

end;

Now we’ll get a notification when a client form is about to be destroyed and the host can delete the icon as in the first way suggested. You can test it with the Report number of icons menu item. 

Want mo’ ?

All you’ve done to add those functionality to the host menu, you could do with any other control like TreeView, Toolbar or anything else in the host. Also, it’s not really necessary, that the client functions are visible. You can add all thinkable services to the host and use it with the clients or vice versa.

The next part will show you how to create implicite functions in the server.

Please add the service “Say.*” to the host’s service list. Complete the CallService event:

Case ….

2:  // Say.*



ShowMessage( PChar(ServiceText)+4 );

That’s all for the host. Now change to the package1 project and add a new form like you added the explorer form and save it as TestUnit. Then add an DynBPLClient component, an Edit control and a button to it. Create an Onclick event for the button and herein call the “Say”-Service. Don’t forget to create and free the form and to send the form to the host:

procedure TTestForm.Button1Click(Sender: TObject);

begin


DynBPLClient1.CallService('Say.' + Edit1.Text, nil, false);

end;

procedure TTestForm.FormCreate(Sender: TObject);

begin


DynBPLClient1.CallService('Add.Form',self,false);

end;

initialization


TestForm := TTestForm.Create(nil);

finalization


TestForm.Free;

In the forms menu you will now find another item “TestForm”. Select it to show the form and click on the button1. As you can see, the main application shows a message with the client’s Edit1.Text.

( All this with 5 lines of code. Simple ? YES !

( Notice, that this example shows how to request implicite service rather than only sending a text variable to the host. If you want to do that, create a TObject derived class with a string member and send it as subject.

Try it yourself

Hope, you’ve got now an idea how to use the Application Framework. Think about your own needs and how you can support them.

Remember, you can have multiple BPL clients on one form, use multiple forms thru one client, have one package with multiple forms / clients or one package for each client…..

Enjoy

Peter Sippel.

