
In this tutorial we will show the new AppObjects technology of AppFW. If you are new to Application Framework, please read the Application Framework Tutorial before you take this tour. 

Application objects are items in a TCollection within the AppFW Components (BPLServer, BPLClient). They are introduced for easy interact between application wide objects of different types, such as forms, menus, GUI managers, config dialogs, commands and so on. With AppObjects we do not have to use service calls for interaction anymore. However, service calls are important to send application wide messages. AppObjects can (should) be published by their owner and can be consumed by other modules or components.

You can find detailed information about AppObjects in the help file.

With AppObjects we will be able to create real reusable modules which can be separately replaced, depending of the user’s needs or taste. 

In this tutorial we will

· create the main container application

· implement several GUI Managers

· write some modules with embedded forms or commands

· see how they all seemlessly work together.

· last but not least – see how we can reuse the container for different applications without changing anything in the program, even use the same exe file (!)

As always, the units are already created, you don’t have to code by yourself. 

A brief look into the AppObjects

TAppObjects (TCollection)

Public Properties

· Items[Index: integer]
Returns the TAppObject item with this Index from the list

Methods

· RegisterObjects
registers all objects in the list

· UnregisterObjects
unregisters all objects in the list

TAppObjec (TCollectionItem)

Public Properties

· Data
Pointer which can be used freely

· ID (read only)
Unique ID of the AppObject, will be automatically created with the object
= BPLObject.Name + 'Obj' + '_' + inttostr(index);
· BPLObject (read only)
is the BPLObject which owns this AppObject
Published properties

Let me first say, that these properties have no restrictions at all, concerning how to assign or use them. This depends fully of the whole implementation and your ideas, as you will see later in this tutorial – it’s an agreement between the different components.

· Caption
Can be used e.g. in menus or toolbars to associate the functionality
· Category
Can be used to group several components, e.g. in toolbars
· Checked
If a GUI manager displays an object link, this property could be useful
· Component
Related component of the AppObject, for example a form (could be anything)
· CustomType
Can be used as “detailed” object type related to the ObjectType property or as an object type which is not in the ObjectType list.
· Enabled
see Checked property
· Glyph
Bitmap for visualisation
· ImageIndex
Index of a bitmap in the image list (instead using a glyph)
· ImageList
can be used instead of the glyph
· MenuPath
can be used to determine the position within a menu tree
· ObjectType
represents the type of this object. Currently predefined types are:

otCommand:

function call
otConfig:

configuration form / dialog
otCustom:

see CustomType property
otDockedForm:
a form which is to be used by a Docking Manager
otEmbeddedForm:
a form which is to be used by a Form Manager
otMainObject:

any application relevant object
otManager:

any object, which manages other objects
otMenu:

a menu tree or item to be merged
otPath:


a path variable, e.g. global inifile name
· Text
another string related to the AppObject. For example the path when using an otPath type

· Visible
see Checked property

Methods

· RegisterObject
registers only this object. All BPL objects will be informed about the registration. There’s a method in the collection, which can register all application objects at once.

· UnregisterObject
unregisters this object. All BPL objects will be informed about the unregistration. The object will be automatically unregistered in the deletion process.
· Execute
calls the object’s OnExecute event, if assigned

Events

The only event of an AppObject is OnExecute (TNotifyEvent). This event is called through the Execute method

TDynBPLBase

Since the collection AppObjects is a member of the BPLBase component, there are some methods related to the AppObjects:

Methods

· function GetNextAppObject(AfterObj: TAppObject; ObjType: TAppObjectType; CustomType: string=''): TAppObject;

Determines the next AppObject after AfterObj (can be nil to get the first one) with a specific ObjectType and CustomType. Useful to get all objects currently registered.


· procedure IterateAppObjects

With this procedure you can iterate all currently registered AppObjects, which is useful at the start procedure of a module. Once the BPL component is created, it gets those notifications automatically (see below).

If a BPL component has the OnAppObjectNotification event assigned, it will be called with noIterate notification.


· function GetAppObjectByID( objID: string ): TAppObject;

Searches for an AppObject with a specific unique ID.

Events

· property OnAppObjectNotification: TOnAppObjectNotifyEvent

TOnAppObjectNotifyEvent = procedure(Sender: TObject; AppObject: TAppObject;    Notification: TAppObjectNotification) of object;

Occurs when something changed with the AppObject

noRegister: Occurs when an AppObject is about to be registered

noUnregister: Occurs when an AppObject is about to be unregistered. The unregistration will be automatically invoked when deleting an AppObject.

noChanged:
A property of the AppObject was changed

noComponentAssigned: A component was assigned to the AppObject

noComponentUnassigning: The assigned component is about to be unassigned

noIterate: You have called the IterateAppObjects procedure

Start with the main container application

The main application is a real plain thing without any GUI components, such as menus or toolbars. It does nothing but loading the modules and support a config dialog. In addition it publishes several AppObjects:

· A Path object with the main ini file path 
depending on the commandline, to make it possible to run different application by reading the configurations in this files

· A Component Object with the main panel to embed forms into it

First of all, we create an application with runtime packages and save the project as MainApp.dpr, the main unit as MainUnit.pas. 

Creating a config dialog unit

Since some of the AppObjects can have a configure dialog and we have no GUI (in the beginning), it must possible to reach these configuration dialogs to build the application. Although we could create a Configuration dialog with a PageControl to embed all dialogs in TabSheets, we will create a simple dialog with a list view, adding the AppObjects as list items. Add a form to the project and save it as configUnit.pas.

On this form we  create a list view with one column and an image list, a Cancel and a OK button. The latter is disabled by default. This will be our modal “select configuration dialog”.

First we need  two public pointers:

…

  public

    BPL: TDynBPLBase;

    ConfigObject: TAppObject;

…

The property BPL will be assigned by the main form before showing this dialog. If it is assigned, we will call the procedure ListObjects which shows all available config objects in the list. The ConfigObject property is the AppObject which has to be executed by the main form.

procedure TConfigForm.FormShow(Sender: TObject);

begin

  ConfigObject := nil;

  lv.Clear;

  ImageList.Clear;

  if Assigned(BPL) then

    ListObjects;

end;

procedure TConfigForm.ListObjects;

var

  obj: TAppObject;

  i: integer;

  li: TListItem;

begin

  obj := BPL.GetNextAppObject(nil,otConfig,'');

  while obj <> nil do

  begin

    i := GetObjectItemIndex(obj);

    if i < 0 then

    begin

      li := lv.Items.Add;

      for i:=0 to lv.Columns.Count-2 do

        li.SubItems.Add('');

    end else

      li := lv.Items[i];

    FillItem(li,obj);

    obj := BPL.GetNextAppObject(obj,otConfig,'');

  end;

end;

In this procedure we determine the available application objects with type otConfig. They are added to the list and adjusted by calling FillItem. Here we assign the given AppObject to the data property of the list item, to have an appropriate pointer to the related Object later on. Besides we add the Image of the AppObject to the listview’s ImageList, if assigned.

procedure TConfigForm.FillItem(item: TListItem; obj: TAppObject);

var

  i: integer;

begin

  // Add new Item

  if item = nil then

  begin

    item := lv.Items.Add;

    for i:=0 to lv.Columns.Count-2 do

      item.SubItems.Add('');

  end;

  item.Data := obj;

  item.Caption := obj.Caption;

  item.ImageIndex := -1;

  // ------------------------------------------------

  // Copy image from AppObject, if assigned

  // ------------------------------------------------

  if item.ImageIndex < 0 then

  begin

    if Assigned(obj.Glyph) and not obj.Glyph.Empty then

      item.ImageIndex := 
ImageList.AddMasked(obj.Glyph, obj.Glyph.TransparentColor)

    else if Assigned(obj.ImageList)

    and (obj.ImageIndex >= 0) then

      item.ImageIndex := ImageList.AddImage(obj.ImageList,obj.ImageIndex);

  end;

end;

Another function is called in the ListObjects procedure to find the index of an application object in the listview:

function TConfigForm.GetObjectItemIndex(obj: TAppObject): integer;

var

  i: integer;

begin

  Result := -1;

  for i:=0 to lv.items.Count-1 do

  begin

    if lv.Items[i].Data = obj then

    begin

      Result := i;

      exit;

    end;

  end;

end;

OK for so far. Now we have to implement reactions for user activities. We must enable the OK button when a listitem is selected, react on double clicks and clicks on the OK button:

procedure TConfigForm.lvChange(Sender: TObject; Item: TListItem;

  Change: TItemChange);

begin

  btnOK.Enabled := lv.SelCount > 0;

end;

procedure TConfigForm.lvDblClick(Sender: TObject);

begin

  btnOK.Click;

end;

procedure TConfigForm.btnOKClick(Sender: TObject);

var

  item: TListItem;

begin

  item := lv.Selected;

  if item.Data <> nil then

    ConfigObject := TAppObject(item.Data);

end;

Implementing the main form functions

The main form consists of just two components: 

· The AppFW  BPL Server

· A TPanel with Alignment alClient and a text
”Use system menu to configure modules...”
This means, that we have at least a system menu item to access the config selection dialog.

Let’s have a short look into the class before going on with the AppObjects:

Since we’ll add a system menu item, we have to define a message number for it. Also, we must create a handler for this message. In addition, we add message handlers for 2 messages we will send by ourselves:

…

  private

  { Private declarations }

    procedure WMSysCommand(var Msg: TMessage); message WM_SYSCOMMAND;

    procedure Start(var WinMsg: TMessage); message WM_USER;

    procedure OpenConfigPackages(var WinMsg: TMessage); message WM_USER+1;…

…

implementation

uses ConfigUnit;

{$R *.dfm}

{$R MainApp_Icons.res}

const

  MID_Config  = WM_USER+10;

  MID_NEWLINE = WM_USER+11;
// for separating the menu item

…

The BPL Server has several application objects (see detailed screenshots below)

Configure Modules

This object is our own config object which will call the BPLServer.ConfigDialog in it’s OnExecute event. To prevent the dialog from being called from a module directly (which could be unloaded in this dialog and if so, be destroyed), we call it via window message queue. Since we defined the procedure OpenConfigPackages as message handler for WM_USER+1, this procedure will be called at the end.

procedure TMainForm.BPLServerAppObjects0Execute(Sender: TObject);

begin

   PostMessage(Handle,WM_USER+1,0,0);

end;

procedure TMainForm.OpenConfigPackages(var WinMsg: TMessage);

begin

  BPLServer.ConfigDialog;

  BPLServer.SaveToIniFile;

end;

To support  GUI representation, we assign “File.Config” to the MenuPath, “Config” to the Category and “Configure Modules” to the Caption.

Main Panel

This object publishes the main panel, which can be used by a form manager as a form host control. We set the ObjectType to otMainObject, the CustomType to MainPanel and finally assign the Panel Control to the Component property.

Exit Command

Because exiting the application is rather a main issue of the application container, we publish this function here as a command type. We set the properties Caption to “Exit”, Category to “Application”, objectType to otCommand. In the OnExecute event we implement the Close:

procedure TMainForm.BPLServerAppObjects2Execute(Sender: TObject);

begin

if MessageDlg('Close Application ?', 

  mtWarning, [mbYes,mbNo], 0) = mrYes  then

    Close;

end;

Application Configuration File

I promised a container application which can run several applications. Therefore we publish the name of application configuration file, which can be either the name of an existing file in the command line or – if no command line is submitted - the application exe name with the extension “.app”. We set the object type to otPath, custom type to AppConfigFile.

You will see later, how we use this AppObject to make our exe a multi-app.

Now we’re ready to implement the main functionality to the main form. In the FormCreate event we add the system menu item + a separator line. Then we determine the application name and assign the path to the related application object.

procedure TMainForm.FormCreate(Sender: TObject);

var

  SysMenu: THandle;

begin

  // Add item to system menu

  SysMenu := GetSystemMenu(Handle, False);

  InsertMenu(SysMenu, Word(-1), MF_SEPARATOR, MID_NEWLINE, '');

  InsertMenu(SysMenu, Word(-1), MF_BYPOSITION, MID_Config, 'Configuration');

  // determine the application config file

  if (ParamCount > 0)

  and FileExists(ParamStr(1)) then

  begin

    Caption := ChangeFileExt(ExtractFileName(ParamStr(1)),'');

    BPLServer.AppObjects.Items[3].Text := ParamStr(1);

  end else

  begin

    Caption := Application.Title;

    BPLServer.AppObjects.Items[3].Text :=

      ChangeFileExt(Application.ExeName,'.app');    

  end;

  BPLServer.IniFileName := BPLServer.AppObjects.Items[3].Text;  

  // register application objects

  BPLServer.AppObjects.RegisterObjects;

  // send start message

  PostMessage(Handle,WM_USER,0,0);

end;

With the PostMessage call we invoke the start routine, which loads the AppFW Configuration and sends a Start message to all BPL clients:

procedure TMainForm.Start(var WinMsg: TMessage);

begin

  BPLServer.LoadFromIniFile;

  BPLServer.CallService('App.Started',nil,true)

end;

Next we write the message handler for our system menu item which calls the OpenConfigDialog, where the config dialog will be opened:

procedure TMainForm.WMSysCommand(var Msg: TMessage);

begin

  if Msg.wParam = MID_CONFIG then

    OpenConfigDialog;

  inherited;

end;

procedure TMainForm.OpenConfigDialog;

var

  obj: TAppObject;

begin

  obj := nil;

  with TConfigForm.Create(self) do

  begin

    try

      BPL := BPLServer;

      if ShowModal = mrOk then

        obj := ConfigObject;

    finally

      Free;

    end;

  end;

  if obj <> nil then

    obj.Execute;

end;

Finally, in the close event, we send a close message to the other BPL clients.

procedure TMainForm.FormClose(Sender: TObject; var Action: TCloseAction);

begin

  BPLServer.CallService('App.End',nil,true);

end;

That’s all so far for the container. As you can see, there ‘s no functionality at all – besides the open config dialog. Now we’re going to implement a form manager to embed forms which are created by other client modules.

Creating the “embedded form manager“

For the module we need a new package. We’ll put it in a subfolder and set the output path in the project options->directories/conditional page to “..\”, so that the compiled packages will reside in the main folder. We will do this for every module we create.

Now we add a new form to the package and add some members to the class:

…

  private

    { Private declarations }


 ActiveForm: TForm;


 _Closing: boolean;

    IniFileName: string;

    LastFormID: string;

    MainPanel: TPanel;

    procedure AddForm( AppObject: TAppObject );

    procedure RemoveForm( AppObject: TAppObject );

    procedure ChangeForm( Form: TForm );

    function GetAppObjectByForm( Form: TForm ): TAppObject;

    procedure LoadFromIniFile;

    procedure SaveToIniFile;

  public

    { Public declarations }

    procedure OnFormCommandClick( Sender: TObject );

…

We write a FormCreate handler, initializing some variables, search for the default AppConfigFile (published by the main app) and call LoadFromIniFile to load our settings.

Then we iterate the AppObjects which are already registered. 

procedure TEFForm.FormCreate(Sender: TObject);

var

  appObj: TAppObject;

begin

  _Closing := false;

  appObj :=

    BPLEmbFormMgr.GetNextAppObject(nil, otPath, 'AppConfigFile');

  if appObj <> nil then

    IniFileName := appObj.Text;

  if IniFileName = '' then

    IniFileName := ChangeFileExt( Application.ExeName, '.ini' );

  LoadFromIniFile;

  ActiveForm := nil;

  BPLEmbFormMgr.IterateAppObjects;

end;

In the FormDestroy event we change the active form to nil, otherwise the main panel would do that.

procedure TEFForm.FormDestroy(Sender: TObject);

begin

  ChangeForm(nil);

end;

Besides, we must have a function to find an AppObject related to a specific form:

function TEFForm.GetAppObjectByForm(Form: TForm): TAppObject;

var

  i: integer;

begin

  Result := nil;

  for i:=0 to BPLEmbFormMgr.AppObjects.Count-1 do

  begin

    if BPLEmbFormMgr.AppObjects.Items[i].Component = Form then

    begin

      Result := BPLEmbFormMgr.AppObjects.Items[i];

      exit;

    end;

  end;

end;

For every registered form we will add an own AppObject later in AddForm procedure and assign the OnFormCommandClick procedure to the OnExecute event. When this event is fired, we will change the active form to the  component of the AppObject, which we will have assigned, too.

procedure TEFForm.OnFormCommandClick(Sender: TObject);

begin

  ChangeForm( TForm(TAppObject(Sender).Data) );

end;

Next we put a TDynBPLClient component onto the form and define just one service: “App.End”. This will be called by the main container app before unloading the modules. Write the ServiceCall event:

procedure TEFForm.BPLEmbFormMgrServiceCall(Sender: TObject;

  ServiceText: String; ServiceNumber: Integer; Subject: TObject;

  out handled: Boolean);

begin

  SaveToIniFile;

  ChangeForm(nil);

  _Closing := true;

end;

Since we listen to just one Service, we can directly do our job. We change the ActiveForm to nil, so the owner module can free it without problems. The we set the closing flag, because we should prevent processing unregistrations / unassignments in the notification event, when app is closing:

procedure TEFForm.BPLEmbFormMgrAppObjectNotification(Sender: TObject;

  AppObject: TAppObject; Notification: TAppObjectNotification);

begin

  if _Closing

  or (AppObject.ObjectType <> otEmbeddedForm)

  or (AppObject.Component = nil)

  or not (AppObject.Component is TForm) then

    exit;

  case Notification of

    noComponentAssigned,

    noIterate,

    noRegister:

      AddForm(AppObject);

    noUnregister,

    noComponentUnassigning:

      RemoveForm(AppObject);

  end;

end;

Quite simple job here. We add forms on registrations / component assignments, remove them when they unregister or unassign the form components. How do we add forms ? Here’s the AddForm procedure:

procedure TEFForm.AddForm(AppObject: TAppObject);

var

  form: TForm;

  cmdObj: TAppObject;

  bmp: TBitmap;

begin

  if MainPanel = nil then

  begin

    cmdObj := BPLEmbFormMgr.GetNextAppObject(nil,otMainObject,'MainPanel');

    if (cmdObj = nil)

    or (cmdObj.Component = nil)

    or not (cmdObj.Component is TPanel) then

      exit;

    MainPanel := TPanel(cmdObj.Component);

  end;

  // -----------------------------------------

  form := TForm(AppObject.Component);

  if GetAppObjectByForm(form) <> nil then

    exit;

  form.BorderStyle := bsNone;

  form.BorderIcons := [];

  form.TabStop := true;

  form.TabOrder := 0;

  form.Align := alClient;

  // Create Command and register it.

  cmdObj := BPLEmbFormMgr.AppObjects.Add;

  cmdObj.Caption := form.Caption;

  cmdObj.ObjectType := otCommand;

  cmdObj.Data := AppObject.Component;

  bmp := TBitmap.Create;

  try

    if AppObject.Glyph.Width > 0 then

      cmdObj.Glyph := AppObject.Glyph

    else if Assigned(AppObject.ImageList)

    and (AppObject.ImageIndex >= 0) then

    begin

      AppObject.ImageList.GetBitmap(AppObject.ImageIndex,bmp);

      cmdObj.Glyph := bmp;

    end else if not Form.Icon.Empty then

    begin



  bmp.Width := Form.Icon.Width;


  
bmp.Height := Form.Icon.Height;

  

bmp.Canvas.Draw( 0, 0, Form.Icon );

      cmdObj.Glyph := bmp;

    end;

  finally

    bmp.Free;

  end;

  cmdObj.Enabled := true;

  cmdObj.Visible := true;

  cmdObj.Checked := false;

  cmdObj.Category := AppObject.Category;

  cmdObj.MenuPath := AppObject.MenuPath;

  cmdObj.Text := AppObject.ID;

  if cmdObj.Category = '' then

    cmdObj.Category := AppObject.BPLObject.Caption;

  cmdObj.OnExecute := OnFormCommandClick;

  cmdObj.RegisterObject;

  // ------------------------------------------------

  // We load the first form or - if we saved a form

  // in a previous session, the last used one.

  // ------------------------------------------------

  if (LastFormID = '')

  and (AppObject.ID = LastFormID) then

  begin

    LastFormID := '';

    ChangeForm(TForm(AppObject.Component));

  end else

    if ActiveForm = nil then

      ChangeForm(TForm(AppObject.Component));

end;

The first step is to find out if the main panel is already assigned. If not, we search for the appropriate AppObject by using a call to GetNextAppObject and – if found - assign it’s component to the MainPanel pointer.

Then we will check if the form is already known and in this case leave the procedure. Otherwise we prepare the form for using it in the main panel.

The next step is to create an AppObject for this form and transfer the properties from the original one, including the glyph or bitmap. The component property is transferred to Data property of the new object to be able to find the appropriate object later.

Finally, we assign the OnExecute event, register the object and change the application’s form to the new one.

This is done in the next routine ChangeForm. We just have to hide thr current form, determine the relevant AppObject for the new form and make this form visible.

procedure TEFForm.ChangeForm(Form: TForm);

var

  cmdObj: TAppObject;

begin

  if ActiveForm <> nil then

  begin

    ActiveForm.Visible := false;

    ActiveForm.Parent := nil;

    cmdObj := GetAppObjectByForm(ActiveForm);

    if cmdObj <> nil then

      cmdObj.Checked := false;

  end;

  ActiveForm := Form;

  if ActiveForm <> nil then

  begin

    ActiveForm.Parent := MainPanel;

    ActiveForm.Visible := true;

    cmdObj := GetAppObjectByForm(Form);

    if cmdObj <> nil then

      cmdObj.Checked := true;

  end;

end;

The next procedure shows how to remove a form and destroy the related AppObject:

procedure TEFForm.RemoveForm(AppObject: TAppObject);

var

  cmdObj, hlpObj: TAppObject;

  form, newForm: TForm;

  newIndex: integer;

begin

  form := TForm(AppObject.Component);

  cmdObj := GetAppObjectByForm(form);

  if cmdObj = nil then

    exit;

  form.Visible := false;

  newIndex := -1;

  newForm := nil;

  if form = ActiveForm then

  begin

    if cmdObj.Index > 0 then

      newIndex := cmdObj.Index-1

    else if cmdObj.Index < BPLEmbFormMgr.AppObjects.Count-1 then

      newIndex := cmdObj.Index+1;

    if newIndex < 0 then

      if BPLEmbFormMgr.AppObjects.Count > 0 then

        newIndex := 0;

    if newIndex >= 0 then

    begin

      hlpObj := BPLEmbFormMgr.AppObjects.Items[newIndex];

      newForm := TForm(hlpObj.Component);

    end;

    ChangeForm(newForm);

  end;

  BPLEmbFormMgr.AppObjects.Delete(cmdObj.Index);

end;

The last task to do is to load and save the settings form or to the ini file:

procedure TEFForm.LoadFromIniFile;

var

  ini: TIniFile;

begin

  ini := TIniFile.Create(IniFileName);

  try

    LastFormID :=

      ini.ReadString('EmbeddedFormManager','LastFormID','');

  finally

    ini.Free;

  end;

end;

procedure TEFForm.SaveToIniFile;

var

  obj: TAppObject;

  ini: TIniFile;

begin

  ini := TIniFile.Create(IniFileName);

  try

    obj := GetAppObjectByForm(ActiveForm);

    if obj = nil then

      LastFormID := ''

    else

      LastFormID := obj.Text;

    ini.WriteString('EmbeddedFormManager','LastFormID',LastFormID);

  finally

    ini.Free;

  end;

end;

That’s all for the embedded form manager. It will show the first registered form, but unfortunately  we can not change to another one. That is, we have no access to the module functions / forms, e.g. through a menu.

This will be our next task. We’ll create a menu manager which populates menu items for specific AppObjects, namely otConfig and otCommand objects.

Creating the Menu Managers

I have created two simple menu managers to show how to use the container app as multi application. The first one uses TMainMenu as GUI component, the other shows all menu items in a TreeView and can activate them by a double click.

In this chapter I’m going to explain the tree manager, because this one uses the multi app configuration file. Please refer to the source of the menu manager to get more details about it’s implementation. However, the main menu manager will be used in the multi applications as well.

First of all, we create a new package and add a new form. We add some components on the form:

1. TTreeView, align to alLeft

2. TSplitter, align to alLeft

3. TDynBPLObject, Services are  “App.Started” and “App.End”

4. TImageList with two images for a) tree nodes and b) items without glyph

5. TPopupMenu with men items: 
- Left, Right, Top and Bottom
- Expand all, Collapse all

We add the following properties as private members:

…

  private

    _Closing: boolean;

    IniFileName: string;

…

We need two events of the main form:

procedure TForm1.FormCreate(Sender: TObject);

var

  appObj: TAppObject;

begin

  _Closing := false;

  appObj :=

    BPLTreeManager.GetNextAppObject(nil, otPath, 'AppConfigFile');

  if appObj <> nil then

    IniFileName := appObj.Text;

  if IniFileName = '' then

    IniFileName := ChangeFileExt( Application.ExeName, '.ini' );

  Splitter.Parent := Application.MainForm;

  tv.Parent := Application.MainForm;

  LoadFromIniFile;

  BPLTreeManager.IterateAppObjects;

end; 

procedure TForm1.FormClose(Sender: TObject; var Action: TCloseAction);

begin

  tv.Parent := self;

  Splitter.Parent := self;

end;

In the FormCreateEvent we search for the AppConfigFile object which we need to load/save our settings later on. Then we set the treeview and the splitter on the mainform by changing their parent property. After calling the LoadFromIniFile procedure we iterate all AppObjects for adding them to the tree view in the AppObjectNotification event:

procedure TForm1.BPLTreeManagerAppObjectNotification(Sender: TObject;

  AppObject: TAppObject; Notification: TAppObjectNotification);

begin

  if _Closing then

    exit;

  case Notification of

    noRegister,

    noIterate:

      begin

        case AppObject.ObjectType of

          otConfig,

          otCommand:

            begin

              AddTreeItem(AppObject);

            end;

        end;

      end;

    noUnregister,

    noComponentUnassigning:

      begin

        case AppObject.ObjectType of

          otConfig,

          otCommand:

            begin

              RemoveObjectLinks(AppObject);

              RemoveEmptyLinks(AppObject);

            end;

        end;

      end;

  end;

end;

For adding new registered components we need three functions:

· AddTreeItem

· GetOrCreateNode

· FindNode

AddTreeItem calls GetOrCreateNode to get or create the appropriate tree node for the new AppObject. GetOrCreateNode itself uses FindNode to determine, if a node already exists. After adding the node, we assign the AppObject as data property and the glyph, if any is assigned. Here is how we do it:

procedure TForm1.AddTreeItem(AppObject: TAppObject);

var

  node, parent: TTreeNode;

begin

  parent := GetOrCreateNode(AppObject);

  if parent <> nil then

  begin

    node := tv.Items.AddChild(parent,AppObject.Caption);

    node.Data := AppObject;

    node.ImageIndex := 1;

    if Assigned(AppObject.Glyph) then

    begin

      node.ImageIndex :=


ImageList.AddMasked(AppObject.Glyph,



AppObject.Glyph.TransparentColor);

    end else

    begin

      if Assigned(AppObject.ImageList)

      and (AppObject.ImageIndex >= 0) then

        node.ImageIndex :=

          ImageList.AddImage(AppObject.ImageList,AppObject.ImageIndex);

    end;

    node.SelectedIndex := node.ImageIndex;

  end;

end;

function TForm1.GetOrCreateNode(AppObject: TAppObject): TTreeNode;

var

  i: integer;

  sl: TStringList;

begin

  Result := FindNode(nil,AppObject.MenuPath);

  if Result <> nil then

    exit;

  sl := TStringList.Create;

  try

    sl.Delimiter := '.';

    sl.DelimitedText := AppObject.MenuPath;

    for i:=0 to sl.Count-1 do

    begin

      Result := tv.Items.AddChild(Result,sl.Strings[i]);

      Result.Text := sl.Strings[i];

    end;

  finally

    sl.Free;

  end;

end;

function TForm1.FindNode(Root: TTreeNode; Path: string): TTreeNode;

var

  i: integer;

  node: TTreeNode;

  caption: string;

begin

  Result := nil;

  i := Pos('.',Path);

  if i = 0 then

    caption := Path

  else

    Caption := LeftStr(Path,i-1);

  if Root = nil then

    node := tv.Items.GetFirstNode

  else

    node := Root.getFirstChild;

  while node <> nil do

  begin

    if node.Text = Caption then

    begin

      if i > 0 then

        Result := FindNode(node,RightStr(Path,Length(Path)-i))

      else

        Result := node;

      if Result <> nil then

        exit;

    end else

      node := Root.GetNextChild(node);

  end;

end;

To remove the items when an AppObject is unregistered, we first remove the node and check then, if any node is unused. This is done by calling the next routines in the AppObjectNotification event:

procedure TForm1.RemoveObjectLinks(AppObject: TAppObject);

var

  i: integer;

  item: TTreeNode;

begin

  if AppObject <> nil then

  begin

    for i:=tv.Items.Count-1 downto 0 do

    begin

      item := tv.Items[i];

      if Pointer(Item.Data) = AppObject then

        tv.Items.Delete(Item);

    end;

  end;

end;

procedure TForm1.RemoveEmptyLinks(AppObject: TAppObject);

var

  i: integer;

  item: TTreeNode;

begin

  if AppObject <> nil then

  begin

    for i:=tv.Items.Count-1 downto 0 do

    begin

      item := tv.Items[i];

      if (item.Data = nil)

      and not item.HasChildren then

        tv.Items.Delete(Item);

    end;

  end;

end;

Now we implement the main functionality for the tree manager, calling the functions:

procedure TForm1.tvDblClick(Sender: TObject);

begin

  if tv.Selected <> nil then

    if tv.Selected.Data <> nil then

      TAppObject(tv.Selected.Data).Execute;

end;

And we don’t forget to manage the alignment of the treeview. Therefore we use the popup menu to call the ChangeAlignment procedure:

procedure TForm1.PopupClick(Sender: TObject);

begin

  case TMenuItem(Sender).Tag of

    1,2,3,4:

      begin

        ChangeAlignment(TMenuItem(Sender).Tag);

        SaveToIniFile;

      end;

    5: tv.FullExpand;

    6: tv.FullCollapse;

  end;

end;

procedure TForm1.ChangeAlignment(Align: integer);

begin

  case Align of

    1: begin

         // Because of creation order:

         if tv.Align in [alTop, alBottom] then

         begin

           Splitter.Align := alLeft;

           tv.Align := alLeft;

         end else

         begin

           tv.Align := alLeft;

           Splitter.Align := alLeft;

         end;

       end;

    2: begin

         tv.Align := alRight;

         Splitter.Align := alRight;

       end;

    3: begin

         tv.Align := alTop;

         Splitter.Align := alTop;

       end;

    4: begin

         tv.Align := alBottom;

         Splitter.Align := alBottom;

       end;

  end;

end;

The last things we have to do is set the closing flag in the ServiceCall event of the BPL component, load and save the settings from and to the ini file:

procedure TForm1.BPLTreeManagerServiceCall(Sender: TObject;

  ServiceText: String; ServiceNumber: Integer; Subject: TObject;

  out handled: Boolean);

begin

  if ServiceNumber = 1 then // App.End

    _Closing := true;

end;

procedure TForm1.LoadFromIniFile;

var

  ini: TIniFile;

begin

  ini := TIniFile.Create(IniFileName);

  try

    ChangeAlignment(ini.ReadInteger('TreeManager','Align',1));

  finally

    ini.Free;

  end;

end;

procedure TForm1.SaveToIniFile;

var

  ini: TIniFile;

  al: integer;

begin

  ini := TIniFile.Create(IniFileName);

  try

    al := 1;

    case tv.Align of

      alRight:    al := 2;

      alTop:      al := 3;

      alBottom:   al := 4;

    end;

    ini.WriteInteger('TreeManager','Align',al);

  finally

    ini.Free;

  end;

end;

OK, hard work so far. At last we can build some test modules to see the result of our effort.

Test modules

We’ll build two test modules to cover the different cases of working with AppObject. It doesn’t matter, what the test modules particularly are doing in their forms. Thus I’ll only describe the AppObject related things here.

Test module 1:

Again we create a new runtime package and add three forms. We add the second and the third form to the uses section of the first one.

…

implementation

uses Unit2, Unit3;

…

Then we drop a TDynBPLClient and a TImageList on the first form. Add any two images to the imagelist, then we need three AppObjects in the BPL client:

This one will publish the Form3 as an embedded form. It uses the first image in the image list as it’s bitmap. Note, that you can not assign Form3 as component at design time. We will do this in the FormCreate event:

procedure TForm1.FormCreate(Sender: TObject);

begin

  BPLTest1.AppObjects.Items[0].Component := Form3;

  BPLTest1.AppObjects.RegisterObjects;

end;

The next AppObject publishes the Form1 itself as embedded form. Also, the image list is used here – with image index 1.

The last one is a command object which opens form2 during the OnExecute event. For this one we load a bitmap into the glyph property (which is just for demo purposes). Here is what we do on execute event:

procedure TForm1.DynBPLClient1AppObjects2Execute(Sender: TObject);

begin

  Form2.Show;

end;

Since you have relationsships between the different units, you must create and free the forms in the right order:

…

initialization

  Form3 := TForm3.Create(nil);

  Form2 := TForm2.Create(nil);

  Form1 := TForm1.Create(nil);

finalization

  Form3.Free;

  Form2.Free;

  Form1.Free;

…

Test module 2

The last one !!!

We create the last runtime package, add a new form and put a TCalendar on it with alClient alignment. Add a TDynBPLClient and create one AppObject:

Quite simple job here, the whole implementation section:

implementation

{$R *.dfm}

procedure TForm1.FormCreate(Sender: TObject);

begin

  BPLTest2.AppObjects.RegisterObjects;

end;

initialization

  Form1 := TForm1.Create(nil);

finalization

  Form1.Free;

end.

Important: Since we already have a unit1 in the first test module, save this unit with another name. Otherwise you won’t get the loaded together into the main app.

Running the application

Run the mainapp.exe. What you see is not very much at this moment. Use the system menu item Configuration to call the configration dialog.Now double click the configure modules entry to call the AppFW configuration dialog. Load all packages we created so far.

voila – your application.

The next time you start the mainapp, the packages will be loaded automatically. Play around with loading / unloading the modules to see the effects. 

Outlook

As you can see, you can choose how to display your manager components. Although they are only demos, they can show you the many possibilities of AppObjects. You can separate your functions from the gui part of your application.

Create gui managers like Outlook bar, Action bar, Tool bar, any 3rd party controls etc. Let the user customize their items and make the changes persistent. Use them in docked forms or directly in the main form. It’s a matter of your ideas.

The Multi Application

Last but not least the multi application gimmick.

As you have seen, we can publish one application configuration file wich can be used by any other module to save settings of used components. If all configuration valiues are in one file, then we have basically the whole application settings there.

We have prepared the mainapp to take one command line parameter as main configuration file. If you start the mainapp.exe with any parameter, you have a completely new application which first of all has to be configured. Start it with another parameter and you have another one and so on. Start it without parameters, the default configuration file will be used.

However, the user won’t start it with a parameter. Thus we have to prepare windows to do this. We will register the extension .app to open through our mainapp. Furthermore, we register the first icon of the mainapp as the default icon of .app files. 

You’ll find a file appfw.reg in the tutorial folder. Change the paths in this file to yours and then run the file, that’s all.

Unfortunately a new item in the New context menu of the explorer has to be registered in different ways, depending of the windows version. So you have to do this by yourself. Another way is to simply create a new text file. Rename it to test1.app and doubleclick it. It will start the mainapp.exe. Load the menu manager and test module 1 with the system menu item configuration. Create another .app file and load the tree manager and Test module 2.

Voila - two different applications. With more and more modules you can configure real different applications with only one container.

OK, that’s all for now. Have fun with the new AppObjects.

Best wishes

Peter Sippel

